

Real level of public investment: How to manage the inflation?

By Ibrahim NGOUHOUE ^a
& Rodrigue Nobosse TCHOFFO ^{b†}

Abstract. When the government collects a supplementary indirect tax on an output, the price of that output increases by consequence. Then, using the resulting revenue for public investments will lead to an under consumption of the total revenue invested. This is due to an inflation that has been created by this mechanism. This paper investigates the determination of the net amount of investment projects taking into account the effect of inflation. We use the computable general equilibrium model to test our hypothesis. As result, we show that, some simulations are needed in order to reach the equilibrium.

Keywords. Government spending, Inflation, Taxes, Investment, Computable general equilibrium.

JEL. C68, E62, H50.

1. Introduction

In economic theory, public investment is considered as a productive investment (Nubukpo, 2007). It generally draws its sources from three modes of financing: either by the non-refundable monetary emission, the domestic or foreign borrowing by the taxes. The last two paths, which are recognized as fiscal policy instruments, are mostly used to finance public investment projects. Tax-driven policy is vital in the sense that it preserves the resources allocated to future generations. However, there has always been a lack of consensus around the existence and operationalization of the tax policy. Indeed, two major thoughts emerged from economic history. Classical school and Keynesian school both agreed that the state intervention through a taxation is harmful to economic activity (Smith, 1779; Ricardo, 1821). For Keynesians, State must intervene not only to carry out its sovereign functions, but also to play a role of regulator (Say, 1805; Keynes, 1936). But to general observation, the state has always participated in the economic action. This is why many works put emphasis on the study of the efficiency of the State's action as an economic agent whose objective is to search for the general interest. In this way, we note for example, studies that have focused on the impact of public spending on growth (Nubukpo, 2007; Rosoiu, 2015; Dion, 2016; Obasikene,

^aFaculty of Economics and Management, University of Dschang, Tunisia.

☎. +237 694 662 260 ✉. ngouhouo@yahoo.fr

^{b†}Faculty of Economics and Management, University of Dschang, Tunisia.

☎. +237 697 542 689 ✉. r_tchoffo@yahoo.fr

Journal of Economics Bibliography

2017; Chu *et al.*, 2018; Elechi & Ibenta, 2019). Government spending can affect growth in two ways, either directly by increasing capital stock through the creation of infrastructure or indirectly by increasing factor productivity through human capital accumulation (Tanzi & Zee, 1997).

Moreover, the imposition of an additional indirect tax on an output results an increase in the value of this good, either luxury or not, as long as the value of the currency remains constant (Ricardo, 1821). In the literature on public spending, an important aspect seems to be commonly ignored.

This is the effect of the inflation created by the imposition of an additional tax on output. Cardenete *et al.*, (2017) have invested substantially in researching the rate of the tax on output that maintains a stable budget deficit by defining the amount of expenditure to be made. But since the tax rate readjustment has long been subject to much criticism, with the result that investors are discouraged when it is revised upward, this approach seems less relevant. For this reason, we focus on the following question: what is the actual level of public investment spending from indirect taxation on production? in other words, how can the loss in the amount of public investment as a result of inflation been determined? A computable general equilibrium model approach derived from Cardenete *et al.*, (2017) will help us to answer this question. Section 2 presents a summary of the works on public expenditure, section 3 is devoted to the methodology, section 4 presents some empirical examples while section 5 concludes.

2. Literature review

The impact of an additional tax depends on the State economic situation. In the expansion phase, the tax will engage the consumers' income without affecting the national wealth. In the recession, the tax will have negative impact on national wealth. Endogenous growth models outside their specificity of integrating external effects are linked to the idea that State has a direct influence on the efficiency of the private sector through its public investments (Nubukpo, 2007). This is why Barro (1991) supports the role of State in the development of infrastructure. He explains in his model that public spending increases productivity both in the consumer sector and in the education sector. Government spending can affect growth in two ways, either directly by increasing capital stock through the creation of infrastructure or indirectly by increasing factor productivity through human capital accumulation (Tanzi & Zee, 1997). In this way, most studies agree that public spending has a positive impact on growth (Nubukpo, 2007; Rosoiu, 2015; Dion, 2016; Obasikene, 2017; Chu *et al.*, 2018; Elechi & Ibenta, 2019). Other studies achieve an opposite result (Barth *et al.*, 1990; Gwartney *et al.*, 1998; Christie, 2012). These authors explain their position to the distortionary effects of high taxes, public borrowing and bureaucratic inefficiency whose effects become predominant in the economic system.

3. Methodology

Investment is a dynamic phenomenon by nature. But its modelling in a static perspective can be simplified by considering it as future demand consumption good by households. We focus here on public investment, the financing of which comes partly from the indirect tax collected on the output of the agriculture, industry and service branches. Here, we mean services by that are both public and private. It is assumed that the government is looking for the appropriate amount to invest in supporting economic activity, therefore he will invest only in the service sector since this is the sector in which he operates the most.

3.1. Description of the model

Following Cardenete *et al.*, (2017), let's consider the following assumptions:

The economy has two factors of production including labour and capital, two consumers, the government, two firms and two goods;

The factors are held by two consumers who sell them to firms and the resulting income is used to finance their consumption;

The value added of each firm, resulting from the transformation of the factors of production, is combined with the intermediate consumption to produce the final output;

Each firm produces only one good; The production, consumption and value-added functions take the Cobb Douglas form with constant returns to scale;

The government has three sources of revenue: the indirect tax on final output, the indirect tax on factors and the direct tax on consumers' income;

Half of the tax collected is transferred to consumers and the other half is used for public investments.

Since investment is an economic phenomenon that is dynamic by nature, its modelling in a static perspective is done by considering it as a consumer good for future i.e. household savings. The latter now have access to private consumer goods and of course to public investment too. Cardenete *et al.*, (2017) describe the behaviour of the investment by

$$INV_j = \lambda_l a_{lj} \quad (1)$$

Where INV_j is the proportion of the good j used for the realization of the investment level λ_l . The level of technology used is given by a_{lj}

The equilibrium system is summarized by¹.

$$\begin{aligned} Y &= TD(P, \omega, P_{N+1}, Y, \lambda_l, E; \mathfrak{S}) \\ S(P, \omega; \mathfrak{S}) &= Z(\omega, Y; \mathfrak{S}) \\ P &= (pva(\omega; \mathfrak{S}).V + P.A).I \\ R(\omega, Y; \mathfrak{S}) - T(P, \omega, Y; \mathfrak{S}) &= P_N.E + D \end{aligned} \quad (2)$$

¹ For more details, see Cardenete *et al.* (2017), chapter 4 pages 71-72

$$I(P_{N+1}, \lambda_i) = S_v(P, \omega; \mathfrak{S}) + D$$

$$P_{N+1} = P_i$$

In this system the government has control over two variables (the level of its expenditures E and the level of the deficit D). He cannot control both at the same time. Therefore, he will endogenize one of the variables and exogenize the other. This is done according to the objective to achieve but it may especially take into consideration the behaviour of the economy. In this context and given the objective set, we must endogenise the public deficit since we want to neutralize it from the amount the government will allocate to its expenditures. The latter must serve to control the level of the deficit.

3.2. Government revenue and expenditure: The budget balance issue

As noted above, government revenue comes from the indirect tax on output of each industry at rate τ . From this rate he draws a TC receipt. A proportion δ of this income is transferred to the different categories of households (rich and poor) at proportions δ_1 and δ_2 respectively. The total amount of transferred income is given by $TR = \theta.TC$. Let D be the value of the budget balance, E_i the government spending in sector i , E the amount of its overall expenditures, and P_i the price of the commodity i , we have:

$$E = \sum_i P_i . E_i \tag{3}$$

$P_i . E_i$ represents the amount of government spending in sector i and

$$D = TC - TR - E = TC - \theta . TC - E = (1 - \theta) . TC - E$$

$$= (1 - \theta) . TC - \sum_i P_i . E_i \tag{4}$$

Thus, if the government decides to invest the amount E_i in sector i in order to balance its budget, ie $D = 0$, we will have:

$$(1 - \theta) . TC - \sum_i P_i . E_i = 0$$

$$\Leftrightarrow (1 - \theta) . TC - P_i . E_i \sum_{j \neq i} P_j . E_j = 0$$

$$\Leftrightarrow E_i = \frac{(1 - \theta) . TC - P_i . E_i \sum_{j \neq i} P_j . E_j}{P_i} \tag{5}$$

To simplify, suppose the government invests all of the revenue E in one sector, such as sector 1 i.e. $E = E_1$. The previous equation (5) becomes:

$$E_i = \frac{(1 - \theta) . TC}{P_i} \tag{6}$$

Journal of Economics Bibliography

Then, leaving the tax τ in the equilibrium system described above not only affects the price P_1 but we show further that TC varies indirectly with the evolution of the price P_1 . Thus, it is not sufficient that equation (6) holds to be sure that the budget deficit D will be null. In general, the increase tax in sector 1 leads to an increase in prices and in turn P_1 too. The mechanism is as follows: when an *ad valorem tax* is imposed on output, entrepreneurs pass it on to the more expensive market price. It ultimately affects, the consumer who will witness a decline in its utility. This is the idea advocated by Ricardo (1821), for whom an increase in the government expenditure financed by an additional tax will always imply an increase in the value of the good, whether luxury or not, as long as the value of the currency remains constant.

We will therefore in general has at the basis $E < (1 - \theta).TC$. This means that a loss of

$(1 - \theta).TC - E$ would be caused by this public investment mechanism.

3.3. Determination of the amount of the budget lost in the public investment mechanism

How can we determine the exact amount of investment lost? this is a key issue that necessitates clarification. The search for this amount of losses caused by the increasing price is fundamentally based on a "simulation algorithm". Everything starts from equation (6) above.

But at the baseline, the equilibrium system presented above is based on a social accounting matrix² in which we suppose the absence of the external agent in the economy. Everything concerns only the internal agents to the economy.

The algorithm of the simulations consists here of making consecutive shocks on equation (6) until one has:

$$E_1 = (1 - \theta).TC \quad (7)$$

- Carries out a first shock then collects the level of the deficit D . This last one with the first shock is in general not null and often equals to $(1 - \theta).TC$. Which implies that $E_1 \neq (1 - \theta).TC$. This is due to the fact that the level of expenditure and the price of the corresponding convenience in this case the services do not vary at the same rate. And depending on the case, if the expenses increase more slowly than the prices, then $E_1 > (1 - \theta).TC$. In the opposite case we will have $E_1 < (1 - \theta).TC$.

- In the second shock, spending and price increase to converge to their equilibrium values. Their rates of increase are falling as a result. If they are zero, we reach the optimum and D equal to zero, otherwise we carry out another additional shock and so on until the desired solution.

NB: in practice, the optimum can be reached from the second shock but more often the third one.

² The latter derives from Tableaux Economiques of François Quesnay

4. Some empirical examples

We present here two examples. The first example, while using the same data defers however from Cardenete *et al.*, (2017) by the fact that it permits us to determine the amount that the government should consider in order to maintain the balance budget and the second one is an application to the Cameroonian economy based on 2016 data. For both, an additional 5% tax is collected on the output of each branch. The government transfers a fraction $\theta= 0.5$ or $\theta= 0.25$ of the total tax collected to the different groups of households. On the one hand we have rich households and on the other hand we have poor households. The government invests accordingly the fraction $(1-\theta)$ in the services. The sharing of the income transferred to households is done in an equal manner i.e $\rho= 0.5$. We note by η the total number of shocks needed to ensure a balanced budget. The difference between the nominal value of public expenditure $(1 - \theta).TC$ and its real value E is indicated by $(1 - \theta).TC - E$. The results are shown in Tables 1 and 2.

The first example shows how to move from a state surplus to a balance budget while the second shows the transition from a state deficit to an equilibrium situation.

The analysis of the results in Table 1 shows that 3 simulations are necessary to make the equilibrium in the budget, whether the government has retained 50% or 75% of the tax collected. On the other hand, the results in Table 2 show that four simulations are needed to restore an equilibrium in the budget.

On the other hand, we notice that the prices of goods are increasing. This increase is mainly due to the 5% tax imposed on the output in each branch. Public spending plays only a marginal role in this increase. Indeed, Table 1 shows for example that during the second simulation, the price has a slight increase of 0.5% which even cancels out during the third simulation. Regarding the impact of public spending on growth, it is clear that they contribute positively to economic growth as theoretically expected. But the impacts can also vary depending on the structure of the economy. We note that the increase in GDP in example 1, which goes from 14.6% to 14.7% when public expenditure increases, does not follow the same trend in example 2. Here there is rather a stable increasing of 10%.

Table 1. Results for the first example

Variables	$(1 - \theta) = 0.5$	$(1 - \theta) = 0.75$
First shock ($\eta = 1$)		
TC	10.869	10.888
P_2	1.157	1.158
E	0	0
D	5.434	8.166
$(1 - \theta).TC$	5.434	8.166
Second shock ($\eta = 2$)		
TC	10.951	11.013

Journal of Economics Bibliography

P_2	1.162	1.166
E	5.457	8.217
D	0.019	0.043
$(1 - \theta).TC$	5.476	8.347
Third shock ($\eta = 3$)		
TC	10.952	11.014
P_2	1.162	1.166
E	5.476	8.260
D	0.000066	0.00022
$(1 - \theta).TC$	5.476	8.260
$(1 - \theta).TC - E$	0.000066	0.00022
GDP	1.146	1.147

Table 2. Results for the second example

Variables	$(1 - \theta) = 0.5$	$(1 - \theta) = 0.75$
First shock ($\eta = 1$)		
TC	1056.93	1057.340
P_3	1.093	1.093
E	0	0
D	528.469	528.670
$(1 - \theta).TC$	528.469	528.670
Second shock ($\eta = 2$)		
TC	1038.379	1038.762
P_3	1.083	1.083
E	523.671	523.868
D	-4.482	-4.487
$(1 - \theta).TC$	519.189	519.381
Third shock ($\eta = 3$)		
TC	1038.537	1038.920
P_3	1.084	1.083
E	519.230	519.422
D	0.038	0.038
$(1 - \theta).TC$	519.268	519.460
Fourth shock ($\eta = 4$)		
TC	1038.535	1038.918
P_3	1.084	1.083
E	519.268	519.460
D	-0.000327	-0.000328
$(1 - \theta).TC$	519.268	519.459
$(1 - \theta).TC - E$	-0.000327	-0.000328
GDP	1.10	1.10

5. Concluding remarks

The objective of this article was to develop a technique for measuring the level of the real public spending with government investments taking into consideration an inflation shock. This mechanism is putting in place

Journal of Economics Bibliography

when the government collects a supplementary indirect tax on output since, it leads to augmenting the price of that output. The result is straightforward on an empirical aspect. Public investment leads to inflation which reduces the real level of these investments. The search for this real value is based on an algorithm of “consecutive simulations” of public expenditures in order to balance the government budget. According to some characteristics specific to an economy, the procedure can start from a situation of budgetary surplus or a deficit situation.

Appendix

This is the GAMS code that has been used to generate output of tables 1 and 2. It could help to understand how we got those results in the main manuscript.

The first code displays the output for Table 1. It derives from Cardenete *et al.*, (2017). The second one is our adaptation from the first one in the Cameroonian economy. Its specificity is that it is based on three sectors: agriculture, industry and services. It uses a Social Accounting Matrix (SAM) of Cameroonian economy for 2016. You should have to copy the latter which is given in table 3 in an (Excel file making sure that it has been named pub.xlsx and the spreadsheet is called Feuil1) and paste it in the main directory where all the gams files are located. Follow this way to get on to the appropriate directory when you have lunched GAMS software (File-view in explorer). You should download the demo version of GAMS at (<http://www.gams.com>)

Code 1 for the first example

```
$title real level of public investment: how to manage inflation?
option          decimals=5;
option          nlp=conopt;
set o           sam accounts / 1*8/
  it(o)         goods /1*3/
  i(it)         goods /1*2/
  k(o)          factors /4*5/
  h(o)          households /6*7/
alias (j,i)
alias(k,l)
alias(o,q);
parameters
e0(h,k)         endowment factor
beta(it,h)      cd utility coefficients
a(i,j)          input-output coefficients
alpha(k,i)      production function coefficients
v(i)            value-added coefficients
inv(i)          investment coefficient
va0(i)          value added
p0(i)           prices for goods
pinv0           price of investment good
w0(k)           prices for factors
y0(i)           total output
pva0(i)         price of value-added
b0(k,i)         flexible factor coefficients
c0(it,h)        individual demand for final consumption
cd0(i)          aggregate demand for final consumption
x0(k,i)         firms factor demand
xd0(k)          aggregate factor demand
iy0(i,j)        intermediate consumption of good i by firm j
gdp0            baseline gdp;
```

table sam(o,q) social accounting matrix entries

	1	2	3	4	5	6	7	8
1	20	50	3			15	12	100
2	30	25	7			30	8	100
3						5	5	10
4	40	10						50
5	10	15						25
6				30	20			50
7				20	5			25
8	100	100	10	50	25	50	25	

;

```
p0(i)           = 1;
w0(k)           = 1;
y0(i)           = sam('8',i);
pva0(i)         = 1;
pinv0           = 1;
c0(it,h)        = sam(it,h);
```

Journal of Economics Bibliography

cd0(i)	= sum(h, c0(i,h));
x0(k,i)	= sam(k,i);
xd0(k)	= sum(i,x0(k,i));
iy0(i,j)	= sam(i,j);
e0(h,k)	= sam(h,k) ;
beta(i,h)	= p0(i)*c0(i,h)/(sum(j,p0(j)*c0(j,h))+pinv0*c0('3',h));
beta('3',h)	= pinv0*c0('3',h)/(sum(j,p0(j)*c0(j,h))+pinv0*c0('3',h));
a(i,j)	= iy0(i,j)/(p0(j)*y0(j));
alpha(k,i)	= w0(k)*x0(k,i)/(sum(l,w0(l)*x0(l,i)));
va0(i)	= sum(k,x0(k,i));
v(i)	= va0(i)/y0(i);
b0(k,i)	= x0(k,i)/(v(i)*y0(i));
inv(i)	= sam(i,'3')/sam('8','3');
gdp0	= sum(k,xd0(k));

display p0,w0,pinv0,y0,pva0,va0,gdp0,c0,cd0,x0,xd0,iy0,b0,e0,inv,a,alpha,v;

parameter dg(i) government demand

```
/1  0
 2  0/;
```

parameter

tau(i)	output tax
m(h)	income tax
t(k)	factor tax
del(h)	lumpsum shares
ro	percentage of transfers to households;

tau(i)=0; m(h)=0; t(k)=0; del(h)=0; ro=0;

variables

p(i)	prices for goods
w(k)	prices for factors
wn(k)	net prices for factors
y(i)	total output
pva(i)	price of value-added
b(k,i)	flexible factor coefficients
c(it,h)	individual demand for final consumption and savings
cd(i)	aggregate demand for final consumption
x(k,i)	firms factor demand
xd(k)	aggregate factor demand
tr	transfers to households
tc	total tax collections
ot	output tax collections
ft	factor tax collections
mt	income tax collections
niv	investment level
pinv	investment price
def	government deficit
gd	government expenditure
iy(i,j)	intermediate consumption
e(h,k)	factor endowment
gdp	gpd variable
z	maximizing dummy ;

equations

vaprice(i)	price index for value added
prices(i)	price formation for goods
priceinv	price of investment
facprices(k)	net and gross factor prices
demand(i)	total demand for goods
housdem(i,h)	households demand for goods
savpriv(h)	savings by households
lab(i)	variable coefficient for labor
cap(i)	variable coefficient for capital

Journal of Economics Bibliography

zdfac(k,i)	firms demand for factors
zfacdem(k)	total demand for factors
govincome	government income
govtrans	government transfers
govsav	savings by government
govdem	government demand
incometax	income tax collections
factortax	factor tax collections
outputtax	output tax collections
eqgoods(i)	equilibrium for goods
eqfactors(k)	equilibrium for factors
savin	macro closure
inter(i,j)	intermediate consumption
eqgdp	gdp income approach
maximand	for objective function;
vaprice(i)..	$pva(i) = e = \text{prod}(k, w(k)**\alpha(k,i)) ;$
prices(i)..	$p(i) = e = (1+\tau(i))*(pva(i)*v(i)+\text{sum}(j,p(j)*a(j,i))) ;$
priceinv..	$pinv = e = \text{sum}(i, p(i)*\text{inv}(i)) ;$
facprices(k)..	$w(k) = e = wn(k)*(1+t(k)) ;$
demand(i)..	$cd(i) = e = \text{sum}(h, c(i,h)) ;$
housdem(i,h)..	$c(i,h) = e = (1-m(h))*\beta(i,h)*(\text{del}(h)*tr+\text{sum}(k, wn(k)*e0(h,k)))/p(i) ;$
savpriv(h)..	$c('3',h) = e = (1-m(h))*\beta('3',h)*(\text{del}(h)*tr+\text{sum}(k, wn(k)*e0(h,k)))/pinv ;$
lab (i)..	$b('4',i) = e = \alpha('4',i)*(w('5')/w('4'))**\alpha('5',i) ;$
cap(i)..	$b('5',i) = e = \alpha('5',i)*(w('4')/w('5'))**\alpha('4',i) ;$
zdfac(k,i)..	$x(k,i) = e = b(k,i)*v(i)*y(i) ;$
zfacdem(k)..	$xd(k) = e = \text{sum}(i, x(k,i)) ;$
govincome..	$tc = e = ot+ft+mt ;$
govtrans..	$tr = e = ro*tc ;$
govsav..	$def = e = tc-tr-gd ;$
govdem..	$gd = e = \text{sum}(i, p(i)*dg(i)) ;$
incometax..	$mt = e = \text{sum}(h, m(h)*(\text{del}(h)*tr+\text{sum}(k, wn(k)*e0(h,k)))) ;$
factortax..	$ft = e = \text{sum}((i,k), t(k)*wn(k)*x(k,i)) ;$
outputtax..	$ot = e = \text{sum}(i, \tau(i)*p(i)*y(i)/(1+\tau(i))) ;$
eqgoods(i)..	$y(i) = e = niv*\text{inv}(i) + dg(i) + cd(i) + \text{sum}(j, a(i,j)*y(j)) ;$
eqfactors(k)..	$xd(k) = e = \text{sum}(h, e0(h,k)) ;$
inter(i,j)..	$iy(i,j) = e = p(i)*a(i,j)*y(j) ;$
savin..	$\text{sum}(i, niv*\text{inv}(i)*p(i)) = e = \text{sum}(h, pinv*c('3',h)) + def ;$
eqgdp..	$gdp = e = \text{sum}(k, xd(k)) + tc-mt ;$
maximand..	$z = e = 1 ;$
model inflation /all/;	
scalar lb lowerbound /1e-4/;	
p.lo(i)=lb; pva.lo(i)=lb; w.lo(k)=lb; wn.lo(k)=lb; pinv.lo=lb;	
y.lo(i)=lb; x.lo(k,i)=lb; xd.lo(k)=lb; c.lo(i,h)=lb; cd.lo(i)=lb; b.lo(k,i)=lb;	
tr.lo=0; tc.lo=0; ot.lo=0; ft.lo=0; mt.lo=0; gd.lo=0; niv.lo=0;	
WN.l('5') = 1; z.fx=1;	
*the numéraire	
wn.fx('4') = 1;	
*initialisation of variables	
p.l(i)=p0(i); pva.l(i)=pva0(i); w.l(k)=w0(k); pinv.l=pinv0; y.l(i)=y0(i);	
x.l(k,i)=x0(k,i); xd.l(k)=xd0(k); c.l(it,h)=c0(it,h); cd.l(i)=cd0(i);	
b.l(k,i)=b0(k,i); iy.l(i,j)=iy0(i,j); niv.l=10; e.l(h,k) = e0(h,k);	
tr.l=0; tc.l=0; ot.l=0; ft.l=0; mt.l=0; def.l=0; gd.l=0; gdp.l = gdp0;	
solve inflation maximizing z using nlp ;	
parameter	
y00(i)	benchmark gross output of i

Journal of Economics Bibliography

```

ny0(i)      benchmark net output of i
pc0(i)      benchmark consumption of i
u0(h)      benchmark utility of h
niv0       benchmark investment level
gdp0       benchmark value of gdp ;

y00(i)      = y.l(i) ;
ny0(i)      = y.l(i)-sum(j, a(i,j)*y.l(j));
pc0(i)      = sum(h, c.l(i,h));
u0(h)      = prod(it, c.l(it,h)**beta(it,h));
niv0       = niv.l;
gdp0       = gdp.l ;

* fiscal policy
tau(i)      = 0.05;
*tau(i)     = 0.1243;
t('4')     = 0.0;
t('5')     = 0.0;
m(h)       = 0.0;
del('6')   = 0.5;          del('7')=1-del('6');
ro         = 0.250 ;
dg('1')    = dg('1')+ 0 ;
dg('2')    = dg('2')+ 0;
*solve under policy

solve inflation maximizing z using nlp ;

dg('2')    = (tc.l -tr.l)/(p.l('2'));

*solve under policy

solve inflation maximizing z using nlp ;

dg('2')    = (tc.l -tr.l)/(p.l('2'));

*solve under policy

solve inflation maximizing z using nlp ;

*write simulation results
parameter
u(h)       simutility
du(h)      utility changes
wag        wages
kap        capital income
pc(i)      simconsumption of good i
prc        private consumption
gdpi       gdp-income
gdpe       gdp-expenditure
fbk        gross capital formation
itax       indirect taxation
sav        savings by households
pubc       public consumption
ny(i)      net output
dny(i)     index for net output of i
dy(i)      index for gross output
dgdpi      index for gdp-income
dinv       index for investment;

u(h)       = prod(it, c.l(it,h)**beta(it,h));
du(h)      = (u(h)/u0(h)-1)*100;
wag        = wn.l('4')*xd.l('4');
kap        = wn.l('5')*xd.l('5');

```

Journal of Economics Bibliography

$pc(i) = \text{sum}(h, c.l(i,h));$
 $prc = \text{sum}(i, p.l(i)*pc(i));$
 $itax = ot.l + ft.l;$
 $gdpi = wag+kap+itax;$
 $fbk = \text{sum}(i, niv.l*inv(i)*p.l(i));$
 $sav = \text{sum}(h, pinv.l*c.l('3',h));$
 $pubc = \text{sum}(i, p.l(i)*dg(i));$
 $gdpe = pubc + prc+ fbk;$
 $ny(i) = y.l(i)-\text{sum}(j, a(i,j)*y.l(j));$
 $dgdpi = gdp.l/gdp0;$

* output indexation

$dny(i) = ny(i)/ny0(i);$
 $dy(i) = y.l(i)/y00(i);$
 $dinv = niv.l/niv0;$

display ro, del, tau, p.l, pinv.l, w.l, dy, tc.l, tr.l, dny, dinv, def.l, dg, du,
fbk, prc, pubc, gdpi, dgdpi, gdpe, wag, kap, itax;

Code 2 for Cameroon economy

\$title real level of public investment: how to manage inflation? for Cameroon
option decimals=5;
option nlp=conopt;

set o sam accounts / agr, ind, ser, iv, lab, cap, rich, poor, tot/

it(o) goods /agr, ind, ser,iv /

i(it) goods /agr, ind, ser /

k(o) factors /lab, cap /

h(o) households /rich, poor /

alias (j,i)

alias(k,l)

alias(o,q);

parameters

$e0(h,k)$ endowment
 $\beta(it,h)$ cd utility coefficients
 $a(i,j)$ input-output coefficients
 $\alpha(k,i)$ production function coefficients
 $v(i)$ value-added coefficients
 $inv(i)$ investment coefficient
 $va0(i)$ value added
 $p0(i)$ prices for goods
 $pinv0$ price of investment good
 $w0(k)$ prices for factors
 $y0(i)$ total output
 $pva0(i)$ price of value-added
 $b0(k,i)$ flexible factor coefficients
 $c0(it,h)$ individual demand for final consumption
 $cd0(i)$ aggregate demand for final consumption
 $x0(k,i)$ firms factor demand
 $xd0(k)$ aggregate factor demand
 $iy0(i,j)$ intermediate consumption of good i by firm j
 $sam(o,q)$ social accounting matrix entries
 $\mu(i)$ technological parameter of value added
 $niv0$ level investment
 $gdp0$ baseline gdp;

*=====importation of data from social accounting matrix=====

Journal of Economics Bibliography

```
$call gdxrw.exe i=pub.xlsx o=pour.gdx par=sam rng=feuil1!a1:j10 rdim=1 cdim=1
$gdxin pour.gdx
$load sam
$gdxin
```

```
display sam;
```

```
*=====initialization and calibration of parameters=====
```

```
p0(i)           = 1;
w0(k)           = 1;
y0(i)           = sam('tot',i);
pva0(i)         = 1;
pinv0           = 1;
c0(it,h)        = sam(it,h);
cd0(i)          = sum(h, c0(i,h));
x0(k,i)         = sam(k,i);
xd0(k)          = sum(i,x0(k,i));
iy0(i,j)        = sam(i,j);
e0(h,k)         = sam(h,k) ;
beta(i,h)       = p0(i)*c0(i,h)/(sum(j,p0(j)*c0(j,h))+pinv0*c0('iv',h));
beta('iv',h)    = pinv0*c0('iv',h)/(sum(j,p0(j)*c0(j,h))+pinv0*c0('iv',h));
a(i,j)          = iy0(i,j)/(p0(j)*y0(j));
alpha(k,i)      = w0(k)*x0(k,i)/(sum(l,w0(l)*x0(l,i)));
va0(i)          = sum(k,x0(k,i));
v(i)            = va0(i)/y0(i);
b0(k,i)         = x0(k,i)/(v(i)*y0(i));
inv(i)          = sam(i,'iv')/sam('tot','iv');
mu(i)           = va0(i)/prod(k, x0(k,i)**alpha(k,i));
gdp0           = sum(k,xd0(k));
niv0            = sam('tot','iv');
```

```
display p0,w0,pinv0,y0, niv0,pva0,va0, gdp0,c0,cd0,x0,xd0,iy0,b0,e0,inv,a,alpha,v,mu;
parameter dg(i) government demand
```

```
  /agr  0
  ind   0
  ser   0/;
```

```
parameter
```

```
tau(i)          output tax
m(h)            income tax
t(k)            factor tax
del(h)          lumpsum shares
ro              percentage of transfers to households;
tau(i)=0; m(h)=0; t(k)=0; del(h)=0; ro=0;
```

```
variables
```

```
p(i)           prices for goods
w(k)           prices for factors
wn(k)          net prices for factors
y(i)           total output
pva(i)         price of value-added
b(k,i)         flexible factor coefficients
c(it,h)        individual demand for final consumption and savings
cd(i)          aggregate demand for final consumption
x(k,i)         firms factor demand
xd(k)          aggregate factor demand
tr             transfers to households
tc             total tax collections
ot             output tax collections
ft            factor tax collections
mt            income tax collections
niv           investment level
```

Journal of Economics Bibliography

pinv	investment price
def	government deficit
gd	government expenditure
iy(i,j)	intermediate consumption
e(h,k)	factor endowment
va(i)	value added for branch i
gdp	gdp at market price income aspect
z	maximizing dummy ;
equations	
vaprice(i)	price index for value added
prices(i)	price formation for goods
priceinv	price of investment
facprices(k)	net and gross factor prices
demand(i)	total demand for goods
housdem(i,h)	households demand for goods
savpriv(h)	savings by households
lab(i)	variable coefficient for labor
cap(i)	variable coefficient for capital
zdfac(k,i)	firms demand for factors
zfacdem(k)	total demand for factors
govincome	government income
govtrans	government transfers
govsav	savings by government
govdem	government demand
incometax	income tax collections
factortax	factor tax collections
outputtax	output tax collections
eqgoods(i)	equilibrium for goods
eqfactors(k)	equilibrium for factors
savin	macro closure
inter(i,j)	intermediate consumption of good i by firm j
eqva(i)	value added for firm i
eqgdp	gdp-income calculation
maximand	for objective function;
vaprice(i)..	$pva(i) = e = \text{prod}(k, w(k)^{\alpha(k,i)})$;
prices(i)..	$p(i) = e = (1 + \tau(i)) * (pva(i) * v(i) + \sum(j, p(j) * a(j,i)))$;
priceinv..	$pinv = e = \sum(i, p(i) * inv(i))$;
facprices(k)..	$w(k) = e = wn(k) * (1 + t(k))$;
demand(i)..	$cd(i) = e = \sum(h, c(i,h))$;
housdem(i,h)..	$c(i,h) = e = (1 - m(h)) * \beta(i,h) * (\text{del}(h) * tr + \sum(k, wn(k) * e(h,k))) / p(i)$;
savpriv(h)..	$c('iv',h) = e = (1 - m(h)) * \beta('iv',h) * (\text{del}(h) * tr + \sum(k, wn(k) * e(h,k))) / pinv$;
lab (i)..	$b('lab',i) = e = \alpha('lab',i) * (w('cap') / w('lab'))^{\alpha('cap',i)}$;
cap(i)..	$b('cap',i) = e = \alpha('cap',i) * (w('lab') / w('cap'))^{\alpha('lab',i)}$;
zdfac(k,i)..	$x(k,i) = e = b(k,i) * v(i) * y(i)$;
zfacdem(k)..	$xd(k) = e = \sum(i, x(k,i))$;
govincome..	$tc = e = ot + ft + mt$;
govtrans..	$tr = e = ro * tc$;
govsav..	$def = e = tc - tr - gd$;
govdem..	$gd = e = \sum(i, p(i) * dg(i))$;
incometax..	$mt = e = \sum(h, m(h) * (\text{del}(h) * tr + \sum(k, wn(k) * e(h,k))))$;
factortax..	$ft = e = \sum((i,k), t(k) * wn(k) * x(k,i))$;
outputtax..	$ot = e = \sum(i, \tau(i) * p(i) * y(i) / (1 + \tau(i)))$;
eqgoods(i)..	$y(i) = e = niv * inv(i) + dg(i) + cd(i) + \sum(j, a(i,j) * y(j))$;
eqfactors(k)..	$xd(k) = e = \sum(h, e(h,k))$;
inter(i,j)..	$iy(i,j) = e = p(i) * a(i,j) * y(j)$;
savin..	$\sum(i, niv * inv(i) * p(i)) = e = \sum(h, pinv * c('iv',h)) + def$;
eqva(i)..	$va(i) = e = \mu(i) * \text{prod}(k, x(k,i)^{\alpha(k,i)})$;
eqgdp..	$gdp = e = \sum(i, va(i)) + tc - mt$;
maximand..	$z = e = 1$;
model	inflation /all/;

Journal of Economics Bibliography

*inflation.iterlim=0;

scalar lb lowerbound /1e-4/;

p.lo(i)=lb; pva.lo(i)=lb; w.lo(k)=lb; wn.lo(k)=lb ; pinv.lo=lb;

y.lo(i)=lb; x.lo(k,i)=lb; xd.lo(k)=lb; va.lo(i)=lb;

c.lo(i,h)=lb; cd.lo(i)=lb; b.lo(k,i)=lb;

tr.lo=0; tc.lo=0; ot.lo=0; ft.lo=0; mt.lo=0; gd.lo=0; niv.lo=0;

wn.fx('lab') = 1;

*initialisation of variables

p.l(i)=p0(i); pva.l(i)=pva0(i); w.l(k)=w0(k); pinv.l=pinv0;

y.l(i)=y0(i); x.l(k,i)=x0(k,i); xd.l(k)=xd0(k);c.l(it,h)=c0(it,h);

cd.l(i)=cd0(i); b.l(k,i)=b0(k,i); iy.l(i,j)=iy0(i,j);

niv.l=10; e.l(h,k) = e0(h,k); va.l(i)=va0(i);

z.fx=1; wn.l('cap') = 1; tr.l=0; tc.l=0; ot.l=0; ft.l=0;

mt.l=0; def.l=0; gd.l=0; gdp.l = gdp0;

solve inflation maximizing z using nlp ;

parameter

y00(i) benchmark gross output of i

ny0(i) benchmark net output of i

pc0(i) benchmark consumption of i

u0(h) benchmark utility of h

gdp0 benchmark value of gdp

niv0 benchmark investment level;

y00(i) = y.l(i) ;

ny0(i) = y.l(i)-sum(j, a(i,j)*y.l(j));

pc0(i) = sum(h, c.l(i,h));

u0(h) = prod(it, c.l(it,h)**beta(it,h));

niv0 = niv.l;

gdp0 = gdp.l ;

* fiscal policy

tau(i) = 0.05;

*tau(i) = 0.1243;

t('lab') = 0.0;

t('cap') = 0.0;

m(h) = 0.0;

del('rich') =0.5; del('poor')=1-del('rich');

ro = 0.50 ;

dg('agr') = dg('agr')+ 0 ;

dg('ind') = dg('ind')+ 0 ;

dg('ser') = dg('ser')+ 0 ;

*solve under policy

solve inflation maximizing z using nlp ;

dg('ser') = (tc.l-tr.l)/(p.l('ser')) ;

solve inflation maximizing z using nlp ;

dg('ser') = (tc.l-tr.l)/(p.l('ser')) ;

solve inflation maximizing z using nlp ;

dg('ser') = (tc.l-tr.l)/(p.l('ser')) ;

solve inflation maximizing z using nlp ;

parameter

Journal of Economics Bibliography

u(h) simutility
 du(h) utility changes
 wag wages
 kap capital income
 pc(i) simconsumption of good i
 prc private consumption
 gdpi gdp-income
 gdpe gdp-expenditure
 fbk gross capital formation
 itax indirect taxation
 sav savings by households
 pubc public consumption
 ny(i) net output
 dny(i) index for net output of i
 dy(i) index for gross output
 dinv index for investment
 dgdpi index for gdp-income
 dx(k,i) index for household payment;

$u(h) = \text{prod}(it, c.l(it,h)**\text{beta}(it,h));$
 $du(h) = (u(h)/u0(h)-1)*100;$
 $wag = \text{wn}.l('lab')*\text{xd}.l('lab');$
 $kap = \text{wn}.l('cap')*\text{xd}.l('cap');$
 $pc(i) = \text{sum}(h, c.l(i,h));$
 $prc = \text{sum}(i, p.l(i)*pc(i));$
 $itax = \text{ot}.l + \text{ft}.l;$
 $gdpi = \text{wag} + \text{kap} + \text{itax};$
 $fbk = \text{sum}(i, \text{niv}.l*\text{inv}(i)*p.l(i));$
 $sav = \text{sum}(h, \text{pinv}.l*c.l('iv',h));$
 $\text{pubc} = \text{sum}(i, p.l(i)*\text{dg}(i));$
 $gdpe = \text{pubc} + \text{prc} + \text{fbk};$
 $ny(i) = y.l(i) - \text{sum}(j, a(i,j)*y.l(j));$

* output indexation

$dny(i) = ny(i)/ny0(i);$
 $dy(i) = y.l(i)/y00(i);$
 $dinv = \text{niv}.l/\text{niv}0;$
 $dx(k,i) = x.l(k,i)/x0(k,i);$
 $dgdpi = \text{gdp}.l/\text{gdp}0;$

display ro , del, dx,tau, p.l, pinv.l, w.l, dy, tr.l,dny,
 dinv,def.l, dg, du, fbk, prc, pubc, gdpi, gdpe, dgdpi,wag, kap, itax;

Table 3. Social Accounting Matrix for Cameroon (SAM 2016)

	AGR	IND	SER	IV	LAB	CAP	RICH	POOR	TOT
AGR	290,476377	920,17953	371,005308	360,761556			875,341576	1582,23565	4400
IND	1413,50707	1660,78384	1394,61223	2633,35889			848,332326	799,405648	8750
SER	1215,06179	1523,68873	1010,68513	1755,87956			1046,5761	448,108699	7000
IV							3229,75	1520,25	4750
LAB	1388,96632	1836,21492	3774,81876						7000
CAP	91,9884397	2809,13298	448,878576						3350
RICH					4021,7148	1978,2852			6000
POOR					2978,2852	1371,7148			4350
TOT	4400	8750	7000	4750	7000	3350	6000	4350	

References

- Barro, R. (1991). Economic growth in a cross section of countries. *Quarterly Journal of Economics*, 106(2), 407-443. doi. [10.2307/2937943](https://doi.org/10.2307/2937943)
- Cardenete, A., Guerra, A., & Sancho, F. (2017). *Applied General Equilibrium: An Introduction Second Edition*. Springer-Verlag. doi. [10.1007/9783-662-54893-6](https://doi.org/10.1007/9783-662-54893-6)
- Christie, T. (2012). The effect of government spending on economic growth: testing the non-linear hypothesis. *Bulletin of Economic Research*, 66(2), 183-204. doi. [10.1111/j.1467-8586.2012.00438.x](https://doi.org/10.1111/j.1467-8586.2012.00438.x)
- Chu, T., Holscher, J., & McCarthy, D. (2018). *The impact of productive and non-productive government expenditure on economic growth: an empirical analysis in high-income versus low-to middle-income economies*. *Empirical Economics* doi. [10.1007/s00181-018-1616-3](https://doi.org/10.1007/s00181-018-1616-3)
- Dione, L. (2016). Composition des dépenses publiques et impacts sur la croissance économique: analyses théoriques et empiriques sur des panels de pays développés, émergents et en voie de développement". *Economies et finances*, Université de Bourgogne, Français. [Retrieved from].
- Elechi, S., & Ibenta, S. (2019). Government expenditure and economic growth: Evidence from the Nigeria economy (1981 - 2016). *Advances in Research*, 19(4), 1-14. doi. [10.9734/AIR/2019/v19i430128](https://doi.org/10.9734/AIR/2019/v19i430128)
- Gwartney, J., Lawson, R., & Holcombe, R. (1998). *The size and functions of government and economic growth*. Washington, DC. [Retrieved from].
- Keynes, J.M. (1936). *The General Theory of Employment, Interest, and Money*. London: Macmillan.
- Nubukpo, K. (2007). Dépenses publiques et croissance des pays de l'union économique et monétaire ouest-africaine (uemoa). *De Boeck Supérieur*, 222(2), 223-250. doi. [10.3917/afco.222.0223](https://doi.org/10.3917/afco.222.0223)
- Obasikene, A. (2017). Government expenditure in Nigeria and its impact on the Nigerian economy, 1986-2014. *Journal on Banking Financial Services and Insurance Research*, 7(11), 1-12.
- Ricardo, D. (1821). *On the Principles of Political Economy and Taxation*. Sraffa, 3rd Edition.
- Rosoiu, I. (2015). Emerging markets queries in finance and business: The impact of the government revenues and expenditures on the economic growth. *Procedia Economics and Finance*, 32: 526-533. doi. [10.1016/S2212-5671\(15\)01428-8](https://doi.org/10.1016/S2212-5671(15)01428-8)
- Say, J.B. (1805). *Traité d'économie politique*. Livre I de la production de la richesse, Paris: Calmann-Levy, 1972.
- Smith, A. (1776). *Recherches sur la nature et les causes de la richesse des nations*. Traduction de Paulette Taieb, PUF Paris 1995, version intégrale de la première édition de 1976.
- Tanzi, V & Zee, H. (1997). Fiscal policy and long-run growth. *IMF Staff Papers*, 44(2), 179-209. doi. [10.2307/3867542](https://doi.org/10.2307/3867542)



Copyrights

Copyright for this article is retained by the author(s), with first publication rights granted to the journal. This is an open-access article distributed under the terms and conditions of the Creative Commons Attribution license (<http://creativecommons.org/licenses/by-nc/4.0>).

